Pensieve header: The 3-Heisenberg Formalism for Θ. Engine modified from pensieve://Projects/BabyDoPeGDO/Solving_to_k=3.nb, px formulas modified from pensieve://Talks/Bonn-2505/. Now in xp order.

$\mathbb{E}[\omega, Q, P\_\epsilon\text{Series}]$ represents ωe^{Q+P}, where ω is a scalar, Q is an ϵ-free quadratic, and $P=\sum_{k=0}^{\$k} P[\![k]\!]\,\epsilon^k$ is a perturbation (it is ill-advised to include ω in P because then it will have log terms).

Scheme: 𝔼_[_] // 𝔼_[_] calls FZip or Zip, which are functionally the same. Zip works by handling the quadratic part and calling PZip for the perturbation-only part. PZip works by iteratively solving the synthesis equation. FZip works by encapsulating coefficients, calling Zip, and back-substituting.

In[∘]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\3H"];
Once[<< KnotTheory`];
```

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[∘]:=
```
$k = 1;
```

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → If[TrueQ@ℰ, ■, ■]];
```

Minor utilities

In[∘]:=
```
CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
(*CoefficientCanonical Form *)
CCF[ℰ_] := Factor[ℰ]; (*CoefficientCanonical Form *)
CF[ℰ_List] := CF /@ ℰ;
CF[ℰ_ϵSeries] := CF /@ Take[ℰ, $k + 1];
CF[ℰ_] := Module[{F}, Expand@Collect[ℰ, (p | x | π | ξ | g)__, F] /. F → CCF];
(*CF[ℰ_] := Module[
   {vs=Cases[ℰ, (p|x|π|ξ)__,∞]},
   Total[(CCF[#[[2]]] (Times@@vs^#[[1]]))& /@ CoefficientRules[Expand[ℰ], vs]]
  ];*)
CF[ℰ_𝔼] := CF /@ ℰ;
CF[𝔼sp___[ℰs___]] := CF /@ 𝔼sp[ℰs];
CF[a_ == b_] := CF[a - b] == 0;
```

In[∘]:=
```
ϵSeries /: S1_ϵSeries ≡ S2_ϵSeries :=
  Length[S1] == Length[S2] ∧ Inner[CF[#1] == CF[#2] &, S1, S2, And];
ϵSeries[] := ϵSeries @@ Table[0, $k + 1];
ϵSeries /: Plus[ss___ϵSeries] := Module[{l = Min[Length /@ {ss}]},
   ϵSeries @@ Total[Take[List @@ #, l] & /@ {ss}]
  ];
(*ϵSeries/: S1_ϵSeries+S2_ϵSeries :=
  ϵSeries@@Table[S1〚k〛+S2〚k〛,{k,Min[Length@S1,Length@S2]}];*)
ϵSeries /: S1_ϵSeries * S2_ϵSeries := ϵSeries @@
   Table[Sum[S1〚j + 1〛 * S2〚k - j + 1〛, {j, 0, k}], {k, 0, Min[Length@S1, Length@S2] - 1}];
ϵSeries /: c_ * S_ϵSeries := (c #) & /@ S;
ϵSeries /: ∂vs__ S_ϵSeries := (s ↦ ∂vs s) /@ S;
```

In[∘]:=
```
Options[PolyPlot1] = {Labeled → False};
PolyPlot1[Δ_, OptionsPattern[]] := Module[{Δ1, crs, m, maxc, minc, s, rect},
  Δ1 = PowerExpand@Expand@Δ;
  rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
  If[Δ1 === 0, Graphics[],
   m = Max[-Exponent[Δ1, T, Min], Exponent[Δ1, T, Max]];
   crs = CoefficientRules[T^m Δ1, {T}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   Graphics[crs /. ({x_} → c_) :→ {
       Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
       Tooltip[Polygon[({x + m - 1 / 2, 0} + #) & /@ rect], c T^(x-m)],
       If[Not@OptionValue[Labeled], {},
        {Black, FontSize → Scaled[1/((m + 1) (6 + maxc))], Text[c T^(x-m), {x + m, 0.5}]}]
      }, AspectRatio → Min[1 / 5 , 1/ Sqrt[m + 1]],
    ImagePadding → None, PlotRangePadding → None]
  ]];
Options[PolyPlot2] = {Labeled → False, Shear → True};
PolyPlot2[θ_, OptionsPattern[]] :=
  Module[{θ1, crs, m1, m2, maxc, minc, s, stone, mat, p},
   θ1 = PowerExpand@Expand@θ;
   If[θ1 === 0, Graphics[{White, Disk[]}],
    If[OptionValue[Shear],
     stone = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
     mat = {{1, -1 / 2}, {0, Sqrt[3] / 2}},
```

```
      stone = {{1, 1}, {-1, 1}, {-1, -1}, {1, -1}} / 2;
      mat = IdentityMatrix[2];
     ];
    m1 = Max[-Exponent[θ1, T₁, Min], Exponent[θ1, T₁, Max]];
    m2 = Max[-Exponent[θ1, T₂, Min], Exponent[θ1, T₂, Max]];
    crs = CoefficientRules[T₁^m1 T₂^m2 θ1, {T₁, T₂}];
    maxc = N@Log@Max@Abs[Last /@ crs];
    minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
    If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
    Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/√2]},*)
      crs /. ({x1_, x2_} → c_) :→ {
         Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
         p = mat.{x1 - m1, x2 - m2};
         Tooltip[Polygon[(p + #) & /@ stone], c T₁^(x1-m1) T₂^(x2-m2)] ,
         If[Not@OptionValue[Labeled], {},
          {Black, FontSize → Scaled[1 / (Max[m1, m2] (10 + maxc))], Text[c T₁^(x1-m1) T₂^(x2-m2), p]}]
        }
     }, ImagePadding → None, PlotRangePadding → None]
   ]];
Options[PolyPlot] = {Labeled → False, ImageSize → Automatic};
PolyPlot[{Δ_, θ_}, opts___Rule] := GraphicsColumn[
   {PolyPlot1[Δ, FilterRules[Join[{opts}, Options[PolyPlot]], Options[PolyPlot1]]],
    PolyPlot2[θ, FilterRules[Join[{opts}, Options[PolyPlot]], Options[PolyPlot2]]]},
   Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
   FilterRules[Join[{opts}, Options[PolyPlot]], Options[GraphicsColumn]]
  ];
```

The Main Program

Variables and their duals:

In[∘]:=
```
{p*, x*, π*, ξ*} = {π, ξ, p, x};
(vs_List)* := (v ↦ v*) /@ vs;
(u_vi__)* := (u*)vi;
```

𝔼 operations:

In[∘]:=
```
𝔼 /: 𝔼[ω1_, Q1_, P1_] ≡ 𝔼[ω2_, Q2_, P2_] := CF[ω1 == ω2] ∧ CF[Q1 == Q2] ∧ (P1 ≡ P2);
𝔼 /: 𝔼[ω1_, Q1_, P1_] 𝔼[ω2_, Q2_, P2_] := 𝔼[ω1 ω2, Q1 + Q2, P1 + P2];
𝔼_{d1_→r1_}[ℰ1s___] ≡ 𝔼_{d2_→r2_}[ℰ2s___] ^:= (d1 == d2) ∧ (r1 == r2) ∧ (𝔼[ℰ1s] ≡ 𝔼[ℰ2s]);
𝔼_{d1_→r1_}[ℰ1s___] 𝔼_{d2_→r2_}[ℰ2s___] ^:= 𝔼_{(d1⋃d2)→(r1⋃r2)} @@ (𝔼[ℰ1s] 𝔼[ℰ2s]);
𝔼_{dr_}[ℰs___]_{$k_} := 𝔼_{dr} @@ 𝔼[ℰs]_{$k};
```

In[∘]:=
```
𝔼_{d1_→r1_}[ℰ1s___] // 𝔼_{d2_→r2_}[ℰ2s___] := Module[{is = r1 ⋂ d2, lvs},
  lvs = Flatten@Table[{x_{ν,$@i}, p_{ν,$@i}}, {ν, 3}, {i, is}];
  𝔼_{(d1⋃Complement[d2,is])→(r2⋃Complement[r1,is])} @@ (Zip_{lvs⋃lvs*}[lvs*.lvs, Times[
      𝔼[ℰ1s] /. Table[(u : (x | p))_{ν_,i} → u_{ν,$@i}, {i, is}],
      𝔼[ℰ2s] /. Table[(u : (ξ | π))_{ν_,i} → u_{ν,$@i}, {i, is}]
     ]])
 ]
```



In[∘]:=
```
Zip_{vs_}[ℱ_, ℰ_] := ((*EchoLabel["EZip called with {vs,ℱ,ℰ} = "][{vs,ℱ,ℰ}];*)
   ⟨ℱ, ℰ⟩ // Zip1_{vs} // EZip23_{vs});
Zip_{vs_}[ℱ_, ℰ_] := ((*EchoLabel["EZip called with {vs,ℱ,ℰ} = "][{vs,ℱ,ℰ}];*)
   ⟨ℱ, ℰ⟩ // Zip1_{vs} // Zip2_{vs} // Zip3_{vs});
```

Getting rid of the quadratic.

Lemma 1. With convergences left to the reader,

$$\left\langle F : \mathcal{E}\, e^{\frac{1}{2}\sum_{i,j\in B} G_{ij} z_i z_j} \right\rangle_B = \det(1-GF)^{-1/2} \left\langle F(1-GF)^{-1} : \mathcal{E} \right\rangle_B$$

In[∘]:=
```
Zip1_{{}} = Identity;
Zip1_{vs_}@⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{ℐ, F, G, u, v},
  (*EchoLabel["EZip1 called with {vs,ℱ,ω,Q,P} = "][{vs,ℱ,ω,Q,P}];*)
  ℐ = IdentityMatrix@Length@vs;
  F = Table[∂_{u,v} ℱ, {u, vs*}, {v, vs*}];
  G = Table[∂_{u,v} Q, {u, vs}, {v, vs}];
  ⟨CF[vs*.F.Inverse[ℐ - G.F].vs* / 2],
   𝔼[CF@PowerExpand@Factor[ω Det[ℐ - G.F]^{-1/2}], CF[Q - vs.G.vs / 2], P]⟩
 ]
```

Getting rid of linear terms.

Lemma 2. $\left\langle F : \mathcal{E}\, e^{\sum_{i\in B} y_i z_i} \right\rangle_B = e^{\frac{1}{2}\sum_{i,j\in B} F_{ij} y_i y_j} \left\langle F :\ \mathcal{E}|_{z_B \to z_B + F y_B} \right\rangle_B.$

In[∘]:=
```
Zip2{} = Identity;
Zip2vs_ @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{F, Y, u, v},
  (*EchoLabel["EZip2 called with {vs,ℱ,ω,Q,P} = "][{vs,ℱ,ω,Q,P}];*)
  F = Table[∂u,v ℱ, {u, vs*}, {v, vs*}];
  Y = Table[∂v Q, {v, vs}];
  CF /@ ⟨ℱ, 𝔼[ω, Q - Y.vs + Y.F.Y / 2, P /. Thread[vs → vs + F.Y]]⟩
 ]
```

Dealing with Feynman diagrams.

Lemma 3. With an extra variable λ, $Z_\lambda := \log[\lambda F : \mathbb{e}^P]_B$ satisfies and is determined by the following PDE / IVP:

$$Z_0 = P \quad \text{and} \quad \partial_\lambda Z_\lambda = \frac{1}{2}\sum_{i,j\in B} F_{ij}\left(\partial_{z_i}\partial_{z_j}Z_\lambda + (\partial_{z_i}Z_\lambda)(\partial_{z_j}Z_\lambda)\right).$$

Note that the power m of λ is at most $k-1+\frac{2k+2}{2}=2k$. We write $Z_\lambda = \sum Z[m]\,\lambda^m$.

In[∘]:=
```
Zip3vs_ @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[{Z, u, v, m, j},
  (*EchoLabel["Zip3 called with {vs,ℱ,ω,Q,P} = "][{vs,ℱ,ω,Q,P}];*)
  Z[0] = P;
  For[m = 0, m < 2 $k + 5, ++m,
   Z[m + 1] = CF[1/(2 (m + 1))
      Sum[∂u*,v* ℱ (∂u,v Z[m] + Sum[(∂u Z[j]) (∂v Z[m - j]), {j, 0, m}]), {u, vs}, {v, vs}]]
  ];
  𝔼[ω, Q, CF[Sum[Z[m], {m, 0, 2 $k + 5}] /. Table[v → 0, {v, vs}]]]
 ]
```

In[∘]:=
```
EZip23vs_ @⟨ℱ_, 𝔼[ω_, Q_, P_]⟩ := Module[
  {nP, nℱ, nQ, j = 0, ps, c, t, rr = {(*release rules*)}},
  nP = Total[
      CoefficientRules[#, vs] /.
       (ps_ → c_) :→ (AppendTo[rr, t[++j] → CF@c]; t[j] (Times @@ vs^ps))
     ] & /@ P;
  nQ = Total[CoefficientRules[Q, vs] /.
     (ps_ → c_) :→ (AppendTo[rr, t[++j] → CF@c]; t[j] (Times @@ vs^ps))];
  nℱ = Total[CoefficientRules[ℱ, vs*] /.
     (ps_ → c_) :→ (AppendTo[rr, t[++j] → CF@c]; t[j] (Times @@ (vs*)^ps))];
  CF[Expand[⟨nℱ, 𝔼[ω, nQ, nP]⟩ // Zip2vs // Zip3vs] /. rr]
 ]
```

In[∘]:=
```
mi_,j_→k_ := 𝔼{i,j}→{k}[1, ∑ν=1^3 (πν,i ξν,j + (πν,i + πν,j) pν,k + (ξν,i + ξν,j) xν,k), εSeries[]]
```

In[∘]:=
```
m_{a,b→c}
```
Out[∘]=

$$\mathbb{E}_{\{a,b\}\to\{c\}}[1,\ p_{1,c}(\pi_{1,a}+\pi_{1,b})+p_{2,c}(\pi_{2,a}+\pi_{2,b})+p_{3,c}(\pi_{3,a}+\pi_{3,b})+\pi_{1,a}\xi_{1,b}+x_{1,c}(\xi_{1,a}+\xi_{1,b})+\pi_{2,a}\xi_{2,b}+x_{2,c}(\xi_{2,a}+\xi_{2,b})+\pi_{3,a}\xi_{3,b}+x_{3,c}(\xi_{3,a}+\xi_{3,b}),\ \epsilon\text{Series}[0,0]]$$

In[∘]:=
```
m_{1,2→1} // m_{1,3→1}
(m_{1,2→1} // m_{1,3→1}) ≡ (m_{2,3→2} // m_{1,2→1})
```
Out[∘]=

$$\mathbb{E}_{\{1,2,3\}\to\{1\}}[1,\ p_{1,1}\pi_{1,1}+p_{1,1}\pi_{1,2}+p_{1,1}\pi_{1,3}+p_{2,1}\pi_{2,1}+p_{2,1}\pi_{2,2}+p_{2,1}\pi_{2,3}+p_{3,1}\pi_{3,1}+p_{3,1}\pi_{3,2}+p_{3,1}\pi_{3,3}+x_{1,1}\xi_{1,1}+\pi_{1,1}\xi_{1,2}+x_{1,1}\xi_{1,2}+\pi_{1,1}\xi_{1,3}+\pi_{1,2}\xi_{1,3}+x_{1,1}\xi_{1,3}+x_{2,1}\xi_{2,1}+\pi_{2,1}\xi_{2,2}+x_{2,1}\xi_{2,2}+\pi_{2,1}\xi_{2,3}+\pi_{2,2}\xi_{2,3}+x_{2,1}\xi_{2,3}+x_{3,1}\xi_{3,1}+\pi_{3,1}\xi_{3,2}+x_{3,1}\xi_{3,2}+\pi_{3,1}\xi_{3,3}+\pi_{3,2}\xi_{3,3}+x_{3,1}\xi_{3,3},\ \epsilon\text{Series}[0,0]]$$

Out[∘]=

True

Some Knot Theory

In[∘]:=
```
Kink_{i_} := CC_3 R_{1,2} // m_{2,3→2} // m_{2,1→i};
OverBar[Kink]_{i_} := CC_3 OverBar[R]_{1,2} // m_{1,3→1} // m_{1,2→i}
```

In[∘]:=
```
RVK[pd_PD] := Module[{n, xs, x, rots, front = {0}, k},
    n = Length@pd; rots = Table[0, {2 n}];
    xs = Cases[pd, x_X :> { Xp[x[[4]], x[[1]]]  PositiveQ@x
                            Xm[x[[2]], x[[1]]]  True      ];
    For[k = 0, k < 2 n, ++k, If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → (xs /. {
              Xp[k + 1, l_] | Xm[l_, k + 1] :> {l, k + 1, 1 - l},
              Xp[l_, k + 1] | Xm[k + 1, l_] :> (++rots[[l]]; {1 - l, k + 1, l})
             })],
      Cases[front, k | -k] /. {k, -k} :> --rots[[k + 1]];
     ]];
    RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:=
```
rot[i_, 0] := 𝔼_{{}→{i}}[1, 0, ϵSeries[]];
rot[i_, n_] := Module[{j}, If[n > 0, rot[i, n - 1] CC_j, rot[i, n + 1] OverBar[CC]_j] // m_{i,j→i}];
```

In[ ]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
 Module[{todo, n, rots, ξ, done, st, cx, ξ1, i, j, k, k1, k2, k3},
  {todo, rots} = List @@ rvk;
  AppendTo[rots, 0];
  n = Length[todo];
  ξ = 𝔼_{}→{0}[1, 0, ϵSeries[]];
  done = {0};
  st = Range[0, 2 n + 1];
  While[{} =!= ($M = todo),
   {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
   {i, j} = List @@ cx;
   ξ1 = Switch[Head[cx],
      Xp, (R_{i,j} OverBar[Kink]_k) // m_{j,k→j},
      Xm, (OverBar[R]_{i,j} Kink_k) // m_{j,k→j}
     ];
   ξ1 = (rot[k, rots[[i]]] ξ1) // m_{k,i→i}; rots[[i]] = 0;
   ξ1 = (ξ1 rot[k, rots[[i + 1]]]) // m_{i,k→i}; rots[[i + 1]] = 0;
   ξ1 = (rot[k, rots[[j]]] ξ1) // m_{k,j→j}; rots[[j]] = 0;
   ξ1 = (ξ1 rot[k, rots[[j + 1]]]) // m_{j,k→j}; rots[[j + 1]] = 0;
   ξ *= ξ1;
   If[MemberQ[done, i], ξ = ξ // m_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
   If[MemberQ[done, i - 1], ξ = ξ // m_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
   If[MemberQ[done, j], ξ = ξ // m_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
   If[MemberQ[done, j - 1], ξ = ξ // m_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
   done = done ⋃ {i - 1, i, j - 1, j};
   todo = DeleteCases[todo, cx]
  ];
  CF /@ (ξ (*/. {x_0→x,y_0→y,a_0→a}*))
 ]
```

ρ_1 testing

The pPush Lagrangian of Talks/Bonn-2505/Rho11AlternativeIntegrands.nb:

```
In[•]:= ℒ[X_{i_,j_}[s_]] := T^{s/2} 𝔼[{
      x_i (p_{i+1} - p_i) + x_j (p_{j+1} - p_j) + (T^s - 1) x_i (p_{i+1} - p_{j+1}),
      0, 0, ϵ (-s/2 + s T^s p_{1+j} x_i + 1/2 s T^s (-1 + T^s) p_{1+i} p_{1+j} x_i^2 -
         1/2 s T^s (-1 + T^s) p_{1+j}^2 x_i^2 - s p_{1+j} x_j - s T^s p_{1+i} p_{1+j} x_i x_j + s T^s p_{1+j}^2 x_i x_j)
    }]
   ℒ[C_{i_}[φ_]] := T^{φ/2} 𝔼[{x_i (p_{i+1} - p_i), ϵ (-φ/2 - φ p_{1+i} x_i)}]
   ℒ[K_] := CF[ℒ /@ Features[K]〚2〛]
   vs[K_] := Join @@ Table[{p_i, x_i}, {i, Features[K]〚1〛}]
```



$p^L = p^H(x, p^+) - xp + xp_+$

Extra Q for integration:
$-xp + xp_+ + x_-p$
$\Rightarrow$ Canonical ordering is xp!

```
In[•]:= R_{i_,j_} = CF[ℒ[X_{i,j}[1]] /. ω_ 𝔼[{Q_, P0_, P1a_, P1b_}] :> 𝔼_{{}→{i,j}}[
          ω, Q, ϵSeries[P0, P1a + P1b /. ϵ → 1]
        ]] /. {p_{i+1} :> p_i, p_{j+1} → p_j} /. {T → T_1, x_α_ :> x_{1,α}, p_α_ :> p_{1,α}};
   R̄_{i_,j_} = CF[ℒ[X_{i,j}[-1]] /. ω_ 𝔼[{Q_, P0_, P1a_, P1b_}] :> 𝔼_{{}→{i,j}}[
          ω, Q, ϵSeries[P0, P1a + P1b /. ϵ → 1]
        ]] /. {p_{i+1} :> p_i, p_{j+1} → p_j} /. {T → T_1, x_α_ :> x_{1,α}, p_α_ :> p_{1,α}};
   CC_{i_} = CF[ℒ[C_i[1]] /. ω_ 𝔼[{Q_, P_}] :> 𝔼_{{}→{i}}[
          ω, Q, ϵSeries[0, P /. ϵ → 1]
        ]] /. {p_{i+1} :> p_i} /. {T → T_1, x_α_ :> x_{1,α}, p_α_ :> p_{1,α}};
   C̄C̄_{i_} = CF[ℒ[C_i[-1]] /. ω_ 𝔼[{Q_, P_}] :> 𝔼_{{}→{i}}[
          ω, Q, ϵSeries[0, P /. ϵ → 1]
        ]] /. {p_{i+1} :> p_i} /. {T → T_1, x_α_ :> x_{1,α}, p_α_ :> p_{1,α}};
```

In[∘]:=
```
$k = 1;
R_{i,j}
R̄_{i,j}
CC_i
CC̄_i
```

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i,j\}}\left[\sqrt{T_1},\ -p_{1,i}\,x_{1,i}+T_1\,p_{1,i}\,x_{1,i}+(1-T_1)\,p_{1,j}\,x_{1,i},\ \epsilon\text{Series}\left[0,\ -\frac{1}{2}+T_1\,p_{1,j}\,x_{1,i}+\frac{1}{2}\,(-1+T_1)\,T_1\,p_{1,i}\,p_{1,j}\,x_{1,i}^2-\frac{1}{2}\,(-1+T_1)\,T_1\,p_{1,j}^2\,x_{1,i}^2-p_{1,j}\,x_{1,j}-T_1\,p_{1,i}\,p_{1,j}\,x_{1,i}\,x_{1,j}+T_1\,p_{1,j}^2\,x_{1,i}\,x_{1,j}\right]\right]$$

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i,j\}}\left[\frac{1}{\sqrt{T_1}},\ -p_{1,i}\,x_{1,i}+\frac{p_{1,i}\,x_{1,i}}{T_1}+\frac{(-1+T_1)\,p_{1,j}\,x_{1,i}}{T_1},\ \epsilon\text{Series}\left[0,\ \frac{1}{2}-\frac{p_{1,j}\,x_{1,i}}{T_1}+\frac{(-1+T_1)\,p_{1,i}\,p_{1,j}\,x_{1,i}^2}{2\,T_1^2}-\frac{(-1+T_1)\,p_{1,j}^2\,x_{1,i}^2}{2\,T_1^2}+p_{1,j}\,x_{1,j}+\frac{p_{1,i}\,p_{1,j}\,x_{1,i}\,x_{1,j}}{T_1}-\frac{p_{1,j}^2\,x_{1,i}\,x_{1,j}}{T_1}\right]\right]$$

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i\}}\left[\sqrt{T_1},\ 0,\ \epsilon\text{Series}\left[0,\ -\frac{1}{2}-p_{1,i}\,x_{1,i}\right]\right]$$

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i\}}\left[\frac{1}{\sqrt{T_1}},\ 0,\ \epsilon\text{Series}\left[0,\ \frac{1}{2}+p_{1,i}\,x_{1,i}\right]\right]$$

In[∘]:=
```
$k = 1; {
  {"CC̄", (CC_1 CC̄_2 // m_{1,2→1}) ≡ 𝔼_{{}→{1}}[1, 0, ϵSeries[]]},
  {"An R1", (CC_3 R_{1,2} // m_{2,3→2} // m_{2,1→1}) ≡ (CC̄_3 R_{1,2} // m_{1,3→1} // m_{1,2→1})},
  {"R2b", (R_{1,2} R̄_{3,4} // m_{1,3→1} // m_{2,4→2}) ≡ 𝔼_{{}→{1,2}}[1, 0, ϵSeries[]]},
  {"R3", (R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3}) ≡ (R_{2,3} R_{4,5} R_{1,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3})}
}
```

Out[∘]=
```
{{CC̄, True}, {An R1, True}, {R2b, True}, {R3, True}}
```

In[∘]:=
```
NewBit[K_] := Module[{Alex = Alexander[K][T_1]},
  Alex^2 Z[K]〚3, 2〛 // Factor]
```

In[∘]:=
```
NewBit /@ AllKnots[{3, 5}]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$\left\{\frac{(-1+T_1)^2\,(1+T_1^2)}{T_1^2},\ 0,\ \frac{(-1+T_1)^2\,(1+T_1^2)\,(2+T_1^2+2\,T_1^4)}{T_1^4},\ \frac{(-1+T_1)^2\,(5-4\,T_1+5\,T_1^2)}{T_1^2}\right\}$$

In papers/APAI, page 7, that was:

$$\left\{\frac{(-1+T_1)^2\,(1+T_1^2)}{T_1^2},\ 0,\ \frac{(-1+T_1)^2\,(1+T_1^2)\,(2+T_1^2+2\,T_1^4)}{T_1^4},\ \frac{(-1+T_1)^2\,(5-4\,T_1+5\,T_1^2)}{T_1^2}\right\}$$

And now θ

The Lagrangian is taken from ThetaAlternativeIntegrands.nb.

In[]:=

$$
\begin{aligned}
&T_3 = T_1\,T_2;\\
&\mathcal{L}_{\mathrm{pPush}}[X_{i\_,j\_}[s\_]] :=\\
&\quad (T_1\,T_2)^s\,\mathbb{E}\Big[\frac{s\,\epsilon}{2} - p_{1,i}\,x_{1,i} + T_1^s\,p_{1,1+i}\,x_{1,i} + (1-T_1^s)\,p_{1,1+j}\,x_{1,i} - p_{1,j}\,x_{1,j} + p_{1,1+j}\,x_{1,j} - p_{2,i}\,x_{2,i} +\\
&\qquad T_2^s\,p_{2,1+i}\,x_{2,i} - s\,\epsilon\,T_2^s\,p_{2,1+i}\,x_{2,i} + (1-T_2^s)\,p_{2,1+j}\,x_{2,i} + s\,\epsilon\,T_1^s\,T_2^s\,p_{1,1+i}\,p_{2,1+j}\,x_{1,i}\,x_{2,i} +\\
&\qquad \frac{s\,\epsilon\,(-1+T_1^s)\,T_2^s\,p_{1,1+j}\,p_{2,1+j}\,x_{1,i}\,x_{2,i}}{-1+T_2^s} + (-1+T_1^s)\,T_2^s\,p_{3,1+j}\,x_{1,i}\,x_{2,i} - p_{2,j}\,x_{2,j} + p_{2,1+j}\,x_{2,j} +\\
&\qquad s\,\epsilon\,p_{2,1+j}\,x_{2,j} - s\,\epsilon\,T_1^s\,p_{1,1+i}\,p_{2,1+j}\,x_{1,i}\,x_{2,j} - \frac{s\,\epsilon\,(-1+T_1^s)\,p_{1,1+j}\,p_{2,1+j}\,x_{1,i}\,x_{2,j}}{-1+T_2^s} + (1-T_1^s)\,p_{3,1+j}\\
&\qquad x_{1,i}\,x_{2,j} + \frac{s\,\epsilon\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{1,1+j}\,p_{2,1+i}\,x_{3,i}}{-1+T_2^s} - \frac{s\,\epsilon\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{1,1+j}\,p_{2,1+j}\,x_{3,i}}{-1+T_2^s} -\\
&\qquad p_{3,i}\,x_{3,i} + (T_1\,T_2)^s\,p_{3,1+i}\,x_{3,i} + s\,\epsilon\,(T_1\,T_2)^s\,p_{3,1+i}\,x_{3,i} + (1-(T_1\,T_2)^s)\,p_{3,1+j}\,x_{3,i} -\\
&\qquad \frac{s\,\epsilon\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{3,1+j}\,x_{3,i}}{-1+T_2^s} - \frac{s\,\epsilon\,T_1^s\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{1,1+i}\,p_{3,1+j}\,x_{1,i}\,x_{3,i}}{-1+T_2^s} +\\
&\qquad \frac{s\,\epsilon\,(-1+T_1^s)\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{1,1+j}\,p_{3,1+j}\,x_{1,i}\,x_{3,i}}{-1+T_2^s} - s\,\epsilon\,(T_1\,T_2)^s\,(-1+T_2^s)\,p_{2,1+j}\,p_{3,1+i}\,x_{2,i}\,x_{3,i} +\\
&\qquad s\,\epsilon\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{2,1+j}\,p_{3,1+j}\,x_{2,i}\,x_{3,i} + 2\,s\,\epsilon\,(T_1\,T_2)^s\,p_{2,1+j}\,p_{3,1+i}\,x_{2,j}\,x_{3,i} +\\
&\qquad \frac{s\,\epsilon\,T_2^s\,(-1+(T_1\,T_2)^s)\,p_{2,1+i}\,p_{3,1+j}\,x_{2,j}\,x_{3,i}}{-1+T_2^s} - \frac{s\,\epsilon\,(-1+2\,T_2^s)\,(-1+(T_1\,T_2)^s)\,p_{2,1+j}\,p_{3,1+j}\,x_{2,j}\,x_{3,i}}{-1+T_2^s} -\\
&\qquad p_{3,j}\,x_{3,j} + p_{3,1+j}\,x_{3,j} + s\,\epsilon\,T_1^s\,p_{1,1+i}\,p_{3,1+j}\,x_{1,i}\,x_{3,j} + \frac{s\,\epsilon\,(-1+T_1^s)\,p_{1,1+j}\,p_{3,1+j}\,x_{1,i}\,x_{3,j}}{-1+T_2^s} -\\
&\qquad s\,\epsilon\,T_2^s\,p_{2,1+i}\,p_{3,1+j}\,x_{2,i}\,x_{3,j} - s\,\epsilon\,p_{2,1+j}\,p_{3,1+j}\,x_{2,i}\,x_{3,j}\Big]
\end{aligned}
$$

In[]:=

$$
\begin{aligned}
&\mathcal{L}_{\mathrm{pPush}}[C_{i\_}[\varphi\_]] :=\\
&\quad (T_1\,T_2)^{\varphi}\,\mathbb{E}\Big[(-p_{1,i}+p_{1,1+i})\,x_{1,i} + (-p_{2,i}+p_{2,1+i})\,x_{2,i} + (-p_{3,i}+p_{3,1+i})\,x_{3,i} + \epsilon\,\varphi\left(\frac{1}{2} + p_{3,1+i}\,x_{3,i}\right)\Big]
\end{aligned}
$$

```
In[∘]:= $k = 1;
R_{i_,j_} = CF[ℒ_pPush[X_{i,j}[1]] /. ω_ 𝔼[L_] :> Module[{B, L1, Q, P0, P1a, P1b},
      L1 = CoefficientRules[B L /. {p_{αβ__} :> B p_{αβ}, x_{αβ__} :> B^-1 x_{αβ}}, {ϵ, B}];
      {ω, Q = {0, 1} /. L1, P0 = {0, 0} /. L1, P1a = {1, 2} /. L1, P1b = {1, 1} /. L1};
      𝔼_{{}→{i,j}}[
        ω, Sum_{ν=1}^{3} ((T_ν - 1) x_{ν,i} (p_{ν,i} - p_{ν,j})),
        ϵSeries[P0, P1a + P1b]
       ] /. {p_{ν_,i+1} :> p_{ν,i}, p_{ν_,j+1} → p_{ν,j}}
     ]];
R̄_{i_,j_} = CF[ℒ_pPush[X_{i,j}[-1]] /. ω_ 𝔼[L_] :> Module[{B, L1, Q, P0, P1a, P1b},
      L1 = CoefficientRules[B L /. {p_{αβ__} :> B p_{αβ}, x_{αβ__} :> B^-1 x_{αβ}}, {ϵ, B}];
      {ω, Q = {0, 1} /. L1, P0 = {0, 0} /. L1, P1a = {1, 2} /. L1, P1b = {1, 1} /. L1};
      𝔼_{{}→{i,j}}[
        ω, Sum_{ν=1}^{3} ((T_ν^-1 - 1) x_{ν,i} (p_{ν,i} - p_{ν,j})),
        ϵSeries[P0, P1a + P1b]
       ] /. {p_{ν_,i+1} :> p_{ν,i}, p_{ν_,j+1} → p_{ν,j}}
     ]];
CC_{i_} = CF[ℒ_pPush[C_i[1]] /. ω_ 𝔼[L_] :> Module[{B, L1, Q, P0, P1a, P1b},
      L1 = CoefficientRules[B L /. {p_{αβ__} :> B p_{αβ}, x_{αβ__} :> B^-1 x_{αβ}}, {ϵ, B}];
      {ω, Q = {0, 1} /. L1, P0 = 0, P1a = 0, P1b = {1, 1} /. L1};
      𝔼_{{}→{i}}[
        ω, 0,
        ϵSeries[P0, P1a + P1b]
       ] /. {p_{ν_,i+1} :> p_{ν,i}}
     ]];
C̄C̄_{i_} = CF[ℒ_pPush[C_i[-1]] /. ω_ 𝔼[L_] :> Module[{B, L1, Q, P0, P1a, P1b},
      L1 = CoefficientRules[B L /. {p_{αβ__} :> B p_{αβ}, x_{αβ__} :> B^-1 x_{αβ}}, {ϵ, B}];
      {ω, Q = {0, 1} /. L1, P0 = 0, P1a = 0, P1b = {1, 1} /. L1};
      𝔼_{{}→{i}}[
        ω, 0,
        ϵSeries[P0, P1a + P1b]
       ] /. {p_{ν_,i+1} :> p_{ν,i}}
     ]];
```

In[∘]:= $\mathbf{R_{i,j}}$

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i,j\}}\Big[T_1 T_2,\ (-1+T_1)\,p_{1,i}\,x_{1,i} + (1-T_1)\,p_{1,j}\,x_{1,i} + (-1+T_2)\,p_{2,i}\,x_{2,i} + (1-T_2)\,p_{2,j}\,x_{2,i} + (-1+T_1T_2)\,p_{3,i}\,x_{3,i} + (1-T_1T_2)\,p_{3,j}\,x_{3,i},$$
$$\epsilon\text{Series}\Big[(-1+T_1)\,T_2\,p_{3,j}\,x_{1,i}\,x_{2,i} + (1-T_1)\,p_{3,j}\,x_{1,i}\,x_{2,j},\ \frac{1}{2} - T_2\,p_{2,i}\,x_{2,i} + T_1T_2\,p_{1,i}\,p_{2,j}\,x_{1,i}\,x_{2,i} +$$
$$\frac{(-1+T_1)\,T_2\,p_{1,j}\,p_{2,j}\,x_{1,i}\,x_{2,i}}{-1+T_2} + p_{2,j}\,x_{2,j} - T_1\,p_{1,i}\,p_{2,j}\,x_{1,i}\,x_{2,j} - \frac{(-1+T_1)\,p_{1,j}\,p_{2,j}\,x_{1,i}\,x_{2,j}}{-1+T_2} +$$
$$\frac{T_2\,(-1+T_1T_2)\,p_{1,j}\,p_{2,i}\,x_{3,i}}{-1+T_2} - \frac{T_2\,(-1+T_1T_2)\,p_{1,j}\,p_{2,j}\,x_{3,i}}{-1+T_2} + T_1T_2\,p_{3,i}\,x_{3,i} - \frac{T_2\,(-1+T_1T_2)\,p_{3,j}\,x_{3,i}}{-1+T_2} -$$
$$\frac{T_1T_2\,(-1+T_1T_2)\,p_{1,i}\,p_{3,j}\,x_{1,i}\,x_{3,i}}{-1+T_2} + \frac{(-1+T_1)\,T_2\,(-1+T_1T_2)\,p_{1,j}\,p_{3,j}\,x_{1,i}\,x_{3,i}}{-1+T_2} -$$
$$T_1\,(-1+T_2)\,T_2\,p_{2,j}\,p_{3,i}\,x_{2,i}\,x_{3,i} + T_2\,(-1+T_1T_2)\,p_{2,j}\,p_{3,j}\,x_{2,i}\,x_{3,i} + 2\,T_1T_2\,p_{2,j}\,p_{3,i}\,x_{2,j}\,x_{3,i} +$$
$$\frac{T_2\,(-1+T_1T_2)\,p_{2,i}\,p_{3,j}\,x_{2,j}\,x_{3,i}}{-1+T_2} - \frac{(-1+2T_2)\,(-1+T_1T_2)\,p_{2,j}\,p_{3,j}\,x_{2,j}\,x_{3,i}}{-1+T_2} +$$
$$T_1\,p_{1,i}\,p_{3,j}\,x_{1,i}\,x_{3,j} + \frac{(-1+T_1)\,p_{1,j}\,p_{3,j}\,x_{1,i}\,x_{3,j}}{-1+T_2} - T_2\,p_{2,i}\,p_{3,j}\,x_{2,i}\,x_{3,j} - p_{2,j}\,p_{3,j}\,x_{2,i}\,x_{3,j}\Big]\Big]$$

In[∘]:= $\mathbf{\overline{R}_{i,j}}$

Out[∘]=

$$\mathbb{E}_{\{\}\to\{i,j\}}\Big[\frac{1}{T_1T_2},\ -\frac{(-1+T_1)\,p_{1,i}\,x_{1,i}}{T_1} + \frac{(-1+T_1)\,p_{1,j}\,x_{1,i}}{T_1} -$$
$$\frac{(-1+T_2)\,p_{2,i}\,x_{2,i}}{T_2} + \frac{(-1+T_2)\,p_{2,j}\,x_{2,i}}{T_2} - \frac{(-1+T_1T_2)\,p_{3,i}\,x_{3,i}}{T_1T_2} + \frac{(-1+T_1T_2)\,p_{3,j}\,x_{3,i}}{T_1T_2},$$
$$\epsilon\text{Series}\Big[-\frac{(-1+T_1)\,p_{3,j}\,x_{1,i}\,x_{2,i}}{T_1T_2} + \frac{(-1+T_1)\,p_{3,j}\,x_{1,i}\,x_{2,j}}{T_1},$$
$$-\frac{1}{2} + \frac{p_{2,i}\,x_{2,i}}{T_2} - \frac{p_{1,i}\,p_{2,j}\,x_{1,i}\,x_{2,i}}{T_1T_2} - \frac{(-1+T_1)\,p_{1,j}\,p_{2,j}\,x_{1,i}\,x_{2,i}}{T_1\,(-1+T_2)} - p_{2,j}\,x_{2,j} + \frac{p_{1,i}\,p_{2,j}\,x_{1,i}\,x_{2,j}}{T_1} +$$
$$\frac{(-1+T_1)\,T_2\,p_{1,j}\,p_{2,j}\,x_{1,i}\,x_{2,j}}{T_1\,(-1+T_2)} - \frac{(-1+T_1T_2)\,p_{1,j}\,p_{2,i}\,x_{3,i}}{T_1\,(-1+T_2)\,T_2} + \frac{(-1+T_1T_2)\,p_{1,j}\,p_{2,j}\,x_{3,i}}{T_1\,(-1+T_2)\,T_2} - \frac{p_{3,i}\,x_{3,i}}{T_1T_2} +$$
$$\frac{(-1+T_1T_2)\,p_{3,j}\,x_{3,i}}{T_1\,(-1+T_2)\,T_2} + \frac{(-1+T_1T_2)\,p_{1,i}\,p_{3,j}\,x_{1,i}\,x_{3,i}}{T_1^2\,(-1+T_2)\,T_2} + \frac{(-1+T_1)\,(-1+T_1T_2)\,p_{1,j}\,p_{3,j}\,x_{1,i}\,x_{3,i}}{T_1^2\,(-1+T_2)\,T_2} -$$
$$\frac{(-1+T_2)\,p_{2,j}\,p_{3,i}\,x_{2,i}\,x_{3,i}}{T_1T_2^2} + \frac{(-1+T_1T_2)\,p_{2,j}\,p_{3,j}\,x_{2,i}\,x_{3,i}}{T_1T_2^2} - \frac{2\,p_{2,j}\,p_{3,i}\,x_{2,j}\,x_{3,i}}{T_1T_2} -$$
$$\frac{(-1+T_1T_2)\,p_{2,i}\,p_{3,j}\,x_{2,j}\,x_{3,i}}{T_1\,(-1+T_2)\,T_2} - \frac{(-2+T_2)\,(-1+T_1T_2)\,p_{2,j}\,p_{3,j}\,x_{2,j}\,x_{3,i}}{T_1\,(-1+T_2)\,T_2} -$$
$$\frac{p_{1,i}\,p_{3,j}\,x_{1,i}\,x_{3,j}}{T_1} - \frac{(-1+T_1)\,T_2\,p_{1,j}\,p_{3,j}\,x_{1,i}\,x_{3,j}}{T_1\,(-1+T_2)} + \frac{p_{2,i}\,p_{3,j}\,x_{2,i}\,x_{3,j}}{T_2} + p_{2,j}\,p_{3,j}\,x_{2,i}\,x_{3,j}\Big]\Big]$$

In[∘]:= $\mathcal{L}_{\mathbf{pPush}}[\mathbf{C_i}[1]]$

Out[∘]=

$$T_1T_2\,\mathbb{E}\Big[(-p_{1,i}+p_{1,1+i})\,x_{1,i} + (-p_{2,i}+p_{2,1+i})\,x_{2,i} + (-p_{3,i}+p_{3,1+i})\,x_{3,i} + \epsilon\left(\frac{1}{2} + p_{3,1+i}\,x_{3,i}\right)\Big]$$

In[∘]:= CC_1 // CF

Out[∘]=

$$\mathbb{E}_{\{\}\to\{1\}}\left[T_1 T_2,\ 0,\ \epsilon\text{Series}\left[0,\ \frac{1}{2}+p_{3,1}\,x_{3,1}\right]\right]$$

In[∘]:= $\overline{CC}_1$ // CF

Out[∘]=

$$\mathbb{E}_{\{\}\to\{1\}}\left[\frac{1}{T_1 T_2},\ 0,\ \epsilon\text{Series}\left[0,\ -\frac{1}{2}-p_{3,1}\,x_{3,1}\right]\right]$$

In[∘]:=

```
$k = 0; {
  {"CC̄", (CC₁ CC̄₂ // m_{1,2→1}) ≡ 𝔼_{{}→{1}}[1, 0, ϵSeries[]]},
  {"An R1", (CC₃ R_{1,2} // m_{2,3→2} // m_{2,1→1}) ≡ (CC̄₃ R_{1,2} // m_{1,3→1} // m_{1,2→1})},
  {"R2b", (R_{1,2} R̄_{3,4} // m_{1,3→1} // m_{2,4→2}) ≡ 𝔼_{{}→{1,2}}[1, 0, ϵSeries[]]},
  {"R3", (R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3}) ≡ (R_{2,3} R_{4,5} R_{1,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3})}
}
```

Out[∘]=

{{CC̄, True}, {An R1, True}, {R2b, True}, {R3, True}}

In[∘]:=

```
$k = 1; HL[(CC₁ CC̄₂ // m_{1,2→1}) ≡ 𝔼_{{}→{1}}[1, 0, ϵSeries[]]]
```

Out[∘]=

True

In[∘]:=

```
$π[ℰ_] := (CF[ℰ /. {p_{αβ__} :→ B^{-1} p_{αβ}, x_{αβ__} :→ B x_{αβ}}] /. B → 0)
```

In[∘]:=

```
$k = 1; HL@$π[ (CC₃ R_{1,2} // m_{2,3→2} // m_{2,1→1}) ≡ (CC̄₃ R_{1,2} // m_{1,3→1} // m_{1,2→1})]
```

Out[∘]=

True

In[∘]:= $R_{1,2}\,\overline{R}_{3,4}$ // $m_{1,3\to1}$ // $m_{2,4\to2}$

Out[∘]=

$$\begin{aligned}&\mathbb{E}_{\{\}\to\{1,2\}}\Big[1,\ 0,\\ &\epsilon\text{Series}\Big[0,\ (-1+T_1)\,(1+T_2)\,p_{3,2}\,x_{1,1}\,x_{2,1}+(-1+T_1)\,T_2\,p_{2,1}\,p_{3,2}\,x_{1,1}\,x_{2,1}^2-(-1+T_1)^2\,T_2^2\,p_{3,2}^2\,x_{1,1}^2\,x_{2,1}^2+\\ &(1-T_1)\,p_{2,1}\,p_{3,2}\,x_{1,1}\,x_{2,1}\,x_{2,2}+(-1+T_1)\,T_2\,p_{2,2}\,p_{3,2}\,x_{1,1}\,x_{2,1}\,x_{2,2}+2\,(-1+T_1)^2\,T_2\,p_{3,2}^2\,x_{1,1}^2\,x_{2,1}\,x_{2,2}+\\ &(1-T_1)\,p_{2,2}\,p_{3,2}\,x_{1,1}\,x_{2,2}^2-(-1+T_1)^2\,p_{3,2}^2\,x_{1,1}^2\,x_{2,2}^2-(-1+T_1)\,(-1+T_2)\,p_{3,1}\,p_{3,2}\,x_{1,1}\,x_{2,1}\,x_{3,1}+\\ &\frac{(-1+T_1)\,(-1+T_1T_2)\,\left(-1+T_2+T_2^2\right)\,p_{3,2}^2\,x_{1,1}\,x_{2,1}\,x_{3,1}}{-1+T_2}+\\ &2\,(-1+T_1)\,p_{3,1}\,p_{3,2}\,x_{1,1}\,x_{2,2}\,x_{3,1}-\frac{(-1+T_1)\,(2+T_2)\,(-1+T_1T_2)\,p_{3,2}^2\,x_{1,1}\,x_{2,2}\,x_{3,1}}{-1+T_2}-\\ &(-1+T_1)\,T_2\,p_{3,2}^2\,x_{1,1}\,x_{2,1}\,x_{3,2}+(-1+T_1)\,p_{3,2}^2\,x_{1,1}\,x_{2,2}\,x_{3,2}\Big]\Big]\end{aligned}$$

In[∘]:=

```
$k = 1; HL@$π[(R_{1,2} R̄_{3,4} // m_{1,3→1} // m_{2,4→2}) ≡ 𝔼_{{}→{1,2}}[1, 0, ϵSeries[]]]
```

Out[∘]=

True

In[∘]:=
```
$k = 1;
HL@$π[(R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3}) ≡ (R_{2,3} R_{4,5} R_{1,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3})]
```

Out[∘]=

True

In[∘]:=
```
Z[Knot[3, 1]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$\mathbb{E}_{\{\}\to\{0\}}\Big[\frac{T_1^2 T_2^2}{\left(1-T_1+T_1^2\right)\left(1-T_2+T_2^2\right)\left(1-T_1 T_2+T_1^2 T_2^2\right)}, 0, \epsilon\text{Series}\Big[0, -\frac{1-T_1+T_1^2-T_2-T_1^3 T_2+T_2^2+T_1^4 T_2^2-T_1 T_2^3-T_1^4 T_2^3+T_1^2 T_2^4-T_1^3 T_2^4+T_1^4 T_2^4}{\left(1-T_1+T_1^2\right)\left(1-T_2+T_2^2\right)\left(1-T_1 T_2+T_1^2 T_2^2\right)} - \Big(\left(-1+T_1\right)\big(3-11 T_2+2 T_1 T_2+5 T_2^2+14 T_1 T_2^2-3 T_1^2 T_2^2-T_2^3-8 T_1 T_2^3-13 T_1^2 T_2^3+6 T_1^3 T_2^3-5 T_2^4+9 T_1 T_2^4+5 T_1^2 T_2^4+5 T_1 T_2^5-7 T_1^2 T_2^5-7 T_1^2 T_2^6+6 T_1^3 T_2^6\big)\, p_{3,0}\, x_{1,0}\, x_{2,0}\Big) \Big/ \Big(\left(-1+T_2\right)\left(1-T_2+T_2^2\right)\left(1-T_1 T_2+T_1^2 T_2^2\right)\Big) - \frac{3\left(-1+T_1\right)^2 T_2\left(1+T_2\right) p_{3,0}^2\, x_{1,0}\, x_{2,0}\, x_{3,0}}{-1+T_2}\Big]\Big]$$

In[∘]:=
```
PolyPlot2[1 - T_1 + T_1^2 - T_2 - T_1^3 T_2 + T_2^2 + T_1^4 T_2^2 - T_1 T_2^3 - T_1^4 T_2^3 + T_1^2 T_2^4 - T_1^3 T_2^4 + T_1^4 T_2^4]
```

Out[∘]=



In[∘]:= **z51 = Z[Knot[5, 1]]**

Out[∘]=

$$\mathbb{E}_{\{\}\to\{0\}}\Big[\frac{T_1^4 T_2^4}{\left(1-T_1+T_1^2-T_1^3+T_1^4\right)\left(1-T_2+T_2^2-T_2^3+T_2^4\right)\left(1-T_1 T_2+T_1^2 T_2^2-T_1^3 T_2^3+T_1^4 T_2^4\right)},\ 0,$$
$$\epsilon\text{Series}\Big[0,\ -\Big(\big(2-2 T_1+2 T_1^2-2 T_1^3+2 T_1^4-2 T_2-2 T_1^5 T_2+2 T_2^2+T_1^2 T_2^2-T_1^3 T_2^2+T_1^4 T_2^2+2 T_1^6 T_2^2-2 T_2^3-T_1^2 T_2^3-T_1^5 T_2^3-2 T_1^7 T_2^3+2 T_2^4+T_1^2 T_2^4+T_1^6 T_2^4+2 T_1^8 T_2^4-2 T_1 T_2^5-T_1^3 T_2^5-T_1^6 T_2^5-2 T_1^8 T_2^5+2 T_1^2 T_2^6+T_1^4 T_2^6-T_1^5 T_2^6+T_1^6 T_2^6+2 T_1^8 T_2^6-2 T_1^3 T_2^7-2 T_1^8 T_2^7+2 T_1^4 T_2^8-2 T_1^5 T_2^8+2 T_1^6 T_2^8-2 T_1^7 T_2^8+2 T_1^8 T_2^8\big)\Big/$$
$$\Big(\left(1-T_1+T_1^2-T_1^3+T_1^4\right)\left(1-T_2+T_2^2-T_2^3+T_2^4\right)\left(1-T_1 T_2+T_1^2 T_2^2-T_1^3 T_2^3+T_1^4 T_2^4\right)\Big)\Big)-$$
$$\Big((-1+T_1)\big(5-19 T_2+4 T_1 T_2+9 T_2^2+26 T_1 T_2^2-9 T_1^2 T_2^2-8 T_2^3-12 T_1 T_2^3-29 T_1^2 T_2^3+10 T_1^3 T_2^3+7 T_2^4+11 T_1 T_2^4+15 T_1^2 T_2^4+28 T_1^3 T_2^4-7 T_1^4 T_2^4-T_2^5-10 T_1 T_2^5-14 T_1^2 T_2^5-18 T_1^3 T_2^5-23 T_1^4 T_2^5+10 T_1^5 T_2^5-9 T_2^6+13 T_1 T_2^6+13 T_1^2 T_2^6+17 T_1^3 T_2^6+11 T_1^4 T_2^6+13 T_1 T_2^7-21 T_1^2 T_2^7-16 T_1^3 T_2^7-10 T_1^4 T_2^7-13 T_1^2 T_2^8+25 T_1^3 T_2^8+9 T_1^4 T_2^8+9 T_1^3 T_2^9-15 T_1^4 T_2^9-11 T_1^4 T_2^{10}+10 T_1^5 T_2^{10}\big)\, p_{3,0}\, x_{1,0}\, x_{2,0}\Big)\Big/$$
$$\Big((-1+T_2)\left(1-T_2+T_2^2-T_2^3+T_2^4\right)\left(1-T_1 T_2+T_1^2 T_2^2-T_1^3 T_2^3+T_1^4 T_2^4\right)\Big)-$$
$$\frac{5\,(-1+T_1)^2\, T_2\,(1+T_2)\, p_{3,0}^2\, x_{1,0}\, x_{2,0}\, x_{3,0}}{-1+T_2}\Big]\Big]$$

In[∘]:= **PolyPlot2**

$$\text{PolyPlot2}\big[2-2 T_1+2 T_1^2-2 T_1^3+2 T_1^4-2 T_2-2 T_1^5 T_2+2 T_2^2+T_1^2 T_2^2-T_1^3 T_2^2+T_1^4 T_2^2+2 T_1^6 T_2^2-2 T_2^3-T_1^2 T_2^3-T_1^5 T_2^3-2 T_1^7 T_2^3+2 T_2^4+T_1^2 T_2^4+T_1^6 T_2^4+2 T_1^8 T_2^4-2 T_1 T_2^5-T_1^3 T_2^5-T_1^6 T_2^5-2 T_1^8 T_2^5+2 T_1^2 T_2^6+T_1^4 T_2^6-T_1^5 T_2^6+T_1^6 T_2^6+2 T_1^8 T_2^6-2 T_1^3 T_2^7-2 T_1^8 T_2^7+2 T_1^4 T_2^8-2 T_1^5 T_2^8+2 T_1^6 T_2^8-2 T_1^7 T_2^8+2 T_1^8 T_2^8\big]$$

Out[∘]=